
1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
24
25
26
27
28
29
30
31
32
33
34
35
36
37
38
39
40
41
42
43
44
45
46
47
48
49
50
51
52
53
54
55
56
57
58
59
60
61
62
63
64
65
66
67
68
69
70
71

$PROBLEM ONE COMPARTMENT ORAL MIXED ORDER ABSORPTION ;DATE = 6/2/04 PROGRAMMER=XXXX
;Units: Time=hr, 
;Concentration=ug/ml, 
;Dose = 100mg and 250mg

$DATA KOKAEST_CRSO.CSV IGNORE=C

$INPUT ID TIME CONC=DV AMT CMT RATE EVID OCC MDV
;RATE = -2 in the data file means that the duration for zero order input will be mentioned 
;in the controlstream

$SUBROUTINE ADVAN2 TRANS2

$PK
OC1=0
IF(OCC.EQ.1)OC1=1
OC2=0
IF(OCC.EQ.2)OC2=1

BOVCL = ETA(7)*OC1+ETA(8)*OC2
BOVV = ETA(9)*OC1+ETA(10)*OC2

POPCL = THETA(1)
POPV = THETA(2)
POPKA = THETA(3)
POPDUR = THETA(4)
POPLAG = THETA(5)
POPFKA = THETA(6)

CL = POPCL*EXP(ETA(1)+BOVCL) ;CL/F in L/hr 
V = POPV*EXP(ETA(2)+BOVV) ;V/F  in L
KA = POPKA*EXP(ETA(3)) ;Absorption rate constant                             
D2 = POPDUR*EXP(ETA(4)) ;Duration of zero order input
ALAG1 = POPLAG*EXP(ETA(5)) ;Lag time after which first order absorption starts
F1 = POPFKA*EXP(ETA(6)) ;Fraction of dose absorbed by first order process
F2 = 1-F1

S2 = V ;Scaling factor

$ERROR
IPRED=F
Y=F+F*ERRCV+ERRSD ;Additive and proportional residual error model

$THETA (0.01,1) ;POPCL/F  
$THETA (0.1,10) ;POPV/F
$THETA (0.1,3) ;POPKA
$THETA (0.1,1) ;POPD2
$THETA (0.1,2) ;POPALAG2
$THETA (0.01,0.3,1) ;POPFka

$OMEGA 0.09 ;BSVCL           
$OMEGA 0.09 ;BSVV          
$OMEGA 0.09 ;BSVKA
$OMEGA 0.04 ;BSVD2
$OMEGA 0.04 ;BSVALAG2
$OMEGA 0.01 ;BSVFka
$OMEGA BLOCK(1) 0.09 ;BOVCL occ=1
$OMEGA BLOCK(1) SAME ;BOVCL occ=2
$OMEGA BLOCK(1) 0.09 ;BOVV occ=1
$OMEGA BLOCK(1) SAME ;BOVV occ=2

$SIGMA
0.002 ;ERRCV            
1 ;ERRSD            

$ESTIMATION METHOD=0 MAXEVAL=9999 PRINT=5 POSTHOC
$COVARIANCE

$TABLE ID TIME DV IPRED
NOPRINT ONEHEADER FILE=KOKAEST_crso.fit

